

An Asynchronous Register Bypass Transformation

Karl Papadantonakis

**Computer Science Department
California Institute of Technology**

caltechCSTR/2003.005

An Asynchronous Register Bypass Transformation

Karl Papadantonakis
Caltech Asynchronous VLSI group
6/25/2003

Acknowledgements

The idea and method of the bypass transformation comes straight from the asynchronous MIPS processor [MiniMIPS] designed by the Caltech Asynchronous VLSI group in 1998, led by Alain Martin. Mika Nyström helped recall and formulate the problem.

Introduction

After a CPU has been decomposed into top-level components such as registers and execution units, each register (or register file) is typically specified as doing a read followed by a write in each cycle: the instruction operands are read, then an operation is performed, and then the result is written. In CHP, this is specified as follows:

```
RegFile ≡
  * [
    RC?rc;
    [rc → RA?ra; RD!regs[ra]
    []¬rc → skip
    ];

    WC?wc;
    [wc → WA?wa; WD?regs[wa]
    []¬wc → skip
    ]
  ]
```

Unfortunately, this is hard to directly implement efficiently, because it appears that the registers must be accessed in alternating read and write phases. Instead, for maximum throughput, we would like for a simultaneous read and write to be possible within a single cycle:

```
Core ≡
  * [(RC?rc; [rc → RA?ra; RD!regs[ra] []¬rc → skip]),
    (WC?wc; [wc → WA?wa; RD?regs[wa] []¬wc → skip])
  ]
```

Fortunately, by introducing a simple control structure around the *Core* process, we can make it implement the *RegFile* specification. Formally, we will *transform* the *RegFile* process into the aggregation of *Core* and a control structure.

As an added benefit, the read-after-write latency is reduced, because a write followed by a read to the same address is handled by the control structure (and the core is not read in this case).

The Bypass Transformation

We begin with a general register[file] specification:

```
RegFile ≡
  * [ RC?rc; [rc → RA?ra; RD!regs[ra] [] ¬rc → skip];
    WC?wc; [wc → WA?wa; WD?regs[wa] [] ¬wc → skip]
  ]
```

The goal is to transform this process into a *Core* process which concurrently reads and writes, together with a collection of very simple processes.

The Sequential Transformation

Insert an initial-send control buffer:

```
WC!false; * [ Wc!(WC?) [] ||
  WC?wc; [wc → WA?wa; WD?regs[wa] [] ¬wc → skip];
  * [ RC?rc; [rc → RA?ra; RD!regs[ra] [] ¬rc → skip];
    WC?wc; [wc → WA?wa; WD?regs[wa] [] ¬wc → skip]
  ]
```

- This is not standard CHP notation: $WC!false$ in this case means channel WC is buffered and initialized with the message **false**.

Rewrite by redefining the loop body:

```
WC!false; * [ Wc!(WC?) [] ||
  * [ WC?wc; [wc → WA?wa; WD?regs[wa] [] ¬wc → skip];
    RC?rc; [rc → RA?ra; RD!regs[ra] [] ¬rc → skip]
  ]
```

Reshuffle the control, using slack-elasticity; this means WD must be buffered (as WC was).

```
WC!false; * [ Wc!(WC?) [] ||
  * [ WC?wc; [wc → WA?wa [] ¬wc → skip],
    RC?rc; [rc → RA?ra [] ¬rc → skip];
    [wc → WD?regs[wa] [] ¬wc → skip];
    [rc → RD!regs[ra] [] ¬rc → skip]
  ]
```

Make a special case for $ra = wa$:

```
WC!false; * [ Wc!(WC?) [] ||
  * [ WC?wc; [wc → WA?wa [] ¬wc → skip],
    RC?rc; [rc → RA?ra [] ¬rc → skip];
    bypass := rc ∧ wc ∧ (wa = ra); readNew := rc ∧ ¬bypass;
    [wc → WD?d [] ¬wc → skip];
    [wc → regs[wa] := d [] ¬wc → skip] “;”
    [bypass → RD!d
     [] readNew → RD!regs[ra]
     [] ¬rc → skip
    ]
  ]
```

Finally, change the “;” to a “,”.

To gain performance, it may be necessary to buffer WA (as WD and WC were buffered).

Bypass Decomposition

Before we can decompose the above, we must make two copies of d (namely $dWrite$ and $dBypass$), so that each copy is used only once. Also we make a variable $dRead$, so that $regs$ is referred to entirely within “one semicolon”:

```

WC!false; *[Wc!(WC?)] ||
*[Wc?wc; [wc → WA?wa []¬wc → skip],
  RC?rc; [rc → RA?ra []¬rc → skip];
  bypass := rc ∧ wc ∧ (wa = ra); readNew := rc ∧ ¬bypass, justWrite := wc ∧ ¬bypass;

  [bypass → WD?d; dWrite := d; dBypass := d
  []justWrite → WD?dWrite
  []¬wc → skip
  ];

  [wc → regs[wa] := dWrite []¬wc → skip],
  [readNew → dRead := regs[ra] []¬readNew → skip];

  [bypass → RD!dBypass
  []readNew → RD!dRead
  []¬rc → skip
  ]]

```

Finally, we replace each block of operations by a corresponding reactive process:

```

Compare ≡
  Compare.WC!false; *[Compare.WC!(WC?)] ||
  *[Compare.WC?wc; [wc → Compare.WA?wa; Core.WA!wa []¬wc → skip],
    Compare.RC?rc; [rc → Compare.RA?ra []¬rc → skip];
    bypass := rc ∧ wc ∧ (wa = ra); readNew := rc ∧ ¬bypass, justWrite := wc ∧ ¬bypass;
    c := makeCommand(rc, wc, bypass, readNew, justWrite);
    WSplit.C!c, Core.C!c, RMerge.C!c,
    [wc → Core.WA!wa []¬wc → skip],
    [readNew → Core.RA!ra []¬readNew → skip]
  ]
WSplit ≡
  *[WSplit.C?c;
    [bypass(c) → WSplit.WD?d; Core.D!d, RMerge.DBypass!d
    []justWrite(c) → Core.D!(WSplit.WD?)
    []¬wc(c) → skip
    ]
  ]
Core ≡
  *[Core.C?c;
    [wc(c) → Core.D?regs[Core.WA?] []¬wc → skip],
    [readNew(c) → RMerge.DRead!regs[Core.RA?] []¬readNew → skip]
  ]
RMerge ≡
  *[RMerge.C?c;
    [bypass(c) → RD!(RMerge.DBypass?)
    []readNew(c) → RD!(RMerge.DRead?)
    []¬rc → skip
    ]
  ]

```

- $WC \equiv Compare.WC$, $WA \equiv Compare.WA$, and $WD \equiv WSplit.WD$.
- $RC \equiv Compare.RC$, $RA \equiv Compare.RA$, and RD comes from $RMerge$.

Block Diagram




Figure 1. Block Diagram.

Dead Control Elimination

We refrain from sending any control signal which would have no effect on the receiver. This saves communication, and eliminates cases which would otherwise need to be considered by the receiver.

Compare \equiv

```

Compare.WC!false; *[Compare.WC!(WC?)] ||
*[Compare.WC?wc; [wc  $\rightarrow$  Compare.WA?wa  $\square$   $\neg$ wc  $\rightarrow$  skip],
  Compare.RC?rc; [rc  $\rightarrow$  Compare.RA?ra  $\square$   $\neg$ rc  $\rightarrow$  skip];
  bypass := rc  $\wedge$  wc  $\wedge$  (wa = ra); readNew := rc  $\wedge$   $\neg$ bypass; justWrite := wc  $\wedge$   $\neg$ bypass;
  c := makeCommand(rc, wc, bypass, readNew, justWrite);
  [wc  $\rightarrow$  WSplit.C!c  $\square$   $\neg$ wc  $\rightarrow$  skip],
  [wc  $\vee$  readNew  $\rightarrow$  Core.C!c  $\square$   $\neg$ wc  $\wedge$   $\neg$ readNew  $\rightarrow$  skip],
  [rc  $\rightarrow$  RMerge.C!c  $\square$   $\neg$ rc  $\rightarrow$  skip],
  [wc  $\rightarrow$  Core.WA!wa  $\square$   $\neg$ wc  $\rightarrow$  skip],
  [readNew  $\rightarrow$  Core.RA!ra  $\square$   $\neg$ readNew  $\rightarrow$  skip]
]
  
```

WSplit \equiv

```

*[WSplit.C?c, WSplit.WD?d;
  [bypass(c)  $\rightarrow$  Core.D!d, RMerge.DBypass!d
   $\square$  justWrite(c)  $\rightarrow$  Core.D!d
]
]
  
```

Core \equiv

```

*[Core.C?c;
  [wc(c)  $\rightarrow$  Core.D?regs[Core.WA?]  $\square$   $\neg$ wc  $\rightarrow$  skip],
  [readNew(c)  $\rightarrow$  RMerge.DRead!regs[Core.RA?]  $\square$   $\neg$ readNew  $\rightarrow$  skip]
]
]
  
```

RMerge \equiv

```

*[RMerge.C?c;
  [bypass(c)  $\rightarrow$  RD!(RMerge.DBypass?)
   $\square$  readNew(c)  $\rightarrow$  RD!(RMerge.DRead?)
]
]
  
```

RegFile \equiv *Compare* \parallel *WSplit* \parallel *Core* \parallel *RMerge*

Bibliography

- [MiniMIPS] **The Design of an Asynchronous MIPS R3000 Microprocessor.** Alain J. Martin, Andrew Lines, Rajit Manohar, Mika Nyström, Paul Penzes, Robert Southworth, Uri Cummings and Tak Kwan Lee. *Proc. 17th Conference on Advanced Research in VLSI, 164-181, IEEE Computer Society Press, 1997.*